

**SOLUCIÓN TP7 PROLOG****1. Representar la información relativa a las siguientes familias:**

*En la primera familia,*

- *el padre es Tomás García Pérez, nacido el 7 de Mayo de 1960, trabaja de profesor y gana 60 euros diarios;*
- *la madre es Ana López Ruiz, nacida el 10 de marzo de 1962, trabaja de médica y gana 90 euros diarios;*
- *el hijo es Juan García López, nacido el 5 de Enero de 1980, estudiante;*
- *la hija es María García López, nacida el 12 de Abril de 1992, estudiante.*

*En la segunda familia,*

- *el padre es José Pérez Ruiz, nacido el 6 de Marzo de 1963, trabaja de pintor y gana 120 euros diarios;*
- *la madre es Luisa Gálvez Pérez, nacida el 12 de Mayo de 1964, trabaja de médica y gana 90 euros diarios;*
- *un hijo es Juan Luis Pérez Pérez, nacido el 5 de Febrero de 1990, estudiante;*
- *una hija es María José Pérez Pérez, nacida el 12 de Junio de 1992, estudiante;*
- *otro hijo es José María Pérez Pérez, nacido el 12 de Julio de 1994, estudiante.*

```
familia(persona([tomas,garcia,perez],
                fecha(7,mayo,1960),
                trabajo(profesor,60)),
        persona([ana,lopez,ruiz],
                fecha(10,marzo,1962),
                trabajo(medica,90)),
        [ persona([juan,garcia,lopez],
                  fecha(5,enero,1990),
                  estudiante),
          persona([maria,garcia,lopez],
                  fecha(12,abril,1992),
                  estudiante) ]).
```

```
familia(persona([jose,perez,ruiz],
                fecha(6,marzo,1963),
                trabajo(pintor,120)),
        persona([luisa,galvez,perez],
                fecha(12,mayo,1964),
                trabajo(medica,90)),
        [ persona([juan_luis,perez,perez],
                  fecha(5,febrero,1990),
                  estudiante),
          persona([maria_jose,perez,perez],
                  fecha(12,junio,1992),
                  estudiante),
          persona([jose_maria,perez,perez],
                  fecha(12,julio,1994),
                  estudiante) ]).
```

**2. Realizar las siguientes consultas:**

- ¿existe familia sin hijos?
- ¿existe familia con un hijo?
- ¿existe familia con dos hijos?
- ¿existe familia con tres hijos?
- ¿existe familia con cuatro hijos.?

```
?- familia(_,_,[ ]).
No
?- familia(_,_,[_]).
No
?- familia(_,_,[_,_]).
Yes
?- familia(_,_,[_,_,_]).
Yes
?- familia(_,_,[_,_,_,_]).
No
```

**3. Buscar los nombres de los padres de familia con tres hijos.**

```
?- familia(persona(NP,_,_),_,[_,_,_]).
NP = [jose, perez, ruiz] ;
No
```

**4. Definir la relación casado(X) que se verifique si X es un hombre casado.**

```
casado(X) :-
    familia(X,_,_).
```

**5. Preguntar por los hombres casados.**

```
?- casado(X).
X = persona([tomas, garcia, perez],
            fecha(7, mayo, 1960),
            trabajo(profesor, 60)) ;
X = persona([jose, perez, ruiz],
            fecha(6, marzo, 1963),
            trabajo(pintor, 120)) ;
No
```

**6. Definir la relación casada(X) que se verifique si X es una mujer casada.**

```
casada(X) :-
    familia(_,X,_).
```

**7. Preguntar por las mujeres casadas.**

```
?- casada(X).
X = persona([ana, lopez, ruiz],
            fecha(10, marzo, 1962),
            trabajo(medica, 90)) ;
X = persona([luisa, galvez, perez],
            fecha(12, mayo, 1964),
            trabajo(medica, 90)) ;
No
```

**8. Determinar el nombre de todas las mujeres casadas que trabajan.**

```
?- casada(persona([N,_,_],_,trabajo(_,_))).
N = ana ;
N = luisa ;
No
```

**9. Definir la relación hijo(X) que se verifique si X figura en alguna lista de hijos.**

```
hijo(X) :-
    familia(_,_,L),
    member(X,L).
```

**10. Preguntar por los hijos.**

```
?- hijo(X).
X = persona([juan,garcia,lopez],fecha(5,enero,1990),estudiante) ;
X = persona([maria,garcia,lopez],fecha(12,abril,1992),estudiante) ;
X = persona([juan_luis,perez,perez],fecha(5,febrero,1990),estudiante) ;
X = persona([maria_jose,perez,perez],fecha(12,junio,1992),estudiante) ;
X = persona([jose_maria,perez,perez],fecha(12,julio,1994),estudiante) ;
No
```

**11. Definir la relación persona(X) que se verifique si X es una persona existente en la base de datos.**

```
persona(X) :-
    casado(X);
    casada(X);
    hijo(X).
```

**12.** Preguntar por los nombres y apellidos de todas las personas existentes en la base de datos.

```
?- persona(persona(X,_,_)).
X = [tomas, garcia, perez] ;
X = [jose, perez, ruiz] ;
X = [ana, lopez, ruiz] ;
X = [luisa, galvez, perez] ;
X = [juan, garcia, lopez] ;
X = [maria, garcia, lopez] ;
X = [juan_luis, perez, perez] ;
X = [maria_jose, perez, perez] ;
X = [jose_maria, perez, perez] ;
No
```

**13.** Determinar todos los estudiantes nacidos antes de 1993.

```
?- persona(persona(X,fecha(_,_,Año),estudiante)), Año < 1993.
X = [juan, garcia, lopez]
Año = 1990 ;
X = [maria, garcia, lopez]
Año = 1992 ;
X = [juan_luis, perez, perez]
Año = 1990 ;
X = [maria_jose, perez, perez]
Año = 1992 ;
No
```

**14.** Definir la relación `fecha_de_nacimiento(X,Y)` de forma que si `X` es una persona, entonces `Y` es su fecha de nacimiento.

```
fecha_de_nacimiento(persona(_,Y,_),Y).
```

**15.** Buscar todos los hijos nacidos en 1992.

```
?- hijo(X), fecha_de_nacimiento(X, fecha(_,_,1992)).
X = persona([maria_jose, perez, perez], fecha(12, junio, 1992), estudiante) ;
No
```

**16.** Definir la relación `sueldo(X,Y)` que se verifique si el sueldo de la persona `X` es `Y`.

```
sueldo(persona(_,_ ,trabajo(_ ,Y)),Y).
sueldo(persona(_,_ ,estudiante),0).
```

**17.** Buscar todas las personas nacidas antes de 1964 cuyo sueldo sea superior a 72 euros diarios.

```
?- persona(X),
   fecha_de_nacimiento(X,fecha(_ ,_ ,Año)),
   Año < 1964,
   sueldo(X,Y),
   Y > 72.
X = persona([jose, perez, ruiz],
            fecha(6, marzo, 1963),
            trabajo(pintor, 120))
Año = 1963
Y = 120 ;
X = persona([ana, lopez, ruiz],
            fecha(10, marzo, 1962),
            trabajo(medica, 90))
Año = 1962
Y = 90 ;
No
```

**18.** Definir la relación `total(L,Y)` de forma que si `L` es una lista de personas, entonces `Y` es la suma de los sueldos de las personas de la lista `L`.

```
total([],0).
total([X|L],Y) :-
  sueldo(X,Y1),
  total(L,Y2),
  Y is Y1 + Y2.
```

**19. Calcular los ingresos totales de cada familia.**

```

?- familia(X,Y,Z),total([X,Y|Z],Total).
X = persona([tomas,garcia,perez],
            fecha(7,mayo,1960),
            trabajo(profesor,60))
Y = persona([ana,lopez,ruiz],
            fecha(10,marzo,1962),
            trabajo(medica,90))
Z = [persona([juan,garcia,lopez],fecha(5,enero,1990),estudiante),
     persona([maria,garcia,lopez],fecha(12,abril,1992),estudiante)]
Total = 150 ;
X = persona([jose,perez,ruiz],
            fecha(6,marzo,1963),
            trabajo(pintor,120))
Y = persona([luisa,galvez,perez],
            fecha(12,mayo,1964),
            trabajo(medica,90))
Z = [persona([juan_luis,perez,perez],fecha(5,febrero,1990),estudiante),
     persona([maria_jose,perez,perez],fecha(12,junio,1982),estudiante)
     persona([jose_maria,perez,perez],fecha(12,julio,1984),estudiante)]
Total = 210 ;
No

```